

Smart switch cuts transformer turn-on current

ROBERT LINDSEY, HANSVEDT EDM, URBANA, IL

Transformer-core saturation can cause inexplicable fuse blowing, system crashes, or premature switch and relay failure. When a core saturates, it loses its inductive characteristics; primary winding current can then reach extremely high values for several ac cycles. Turning on a transformer may seem fundamental, but in some power-supply designs and control applications, it can be a game of Russian roulette. Because transformers remain polarized when turned off, saturation occurrence is a function of the polarity and phase angle of the ac cycle when you switch the circuit on and off. The smart-switch circuit in **Figure 1** eliminates saturation, improves relay reliability, and provides a tool for determining transformer and relay performance.

The circuit goes beyond typical configurations using zero-crossing or peak-switching relays, by using the polarity of the ac cycle, known phase angles, and soft-starting techniques. **Figure 2** shows that the primary turn-on current of a 220-VA transformer can be disastrous when you use a zero-crossing relay. Trace R1 shows 46A peak with a saturated

core. Trace 1 shows only a few amps with use of the smart-switch circuit. This large difference in current demonstrates the value of the smart switch in controlling transformer magnetization. Switching on during a positive half cycle and off during a negative half cycle or vice versa prevents most core saturation.

Peak switching of the ac voltage during turn-on and -off further reduces the susceptibility to core saturation, regardless of ac polarity. This reduction is an important consideration in the event of an uncontrolled power outage. **Figure 3**, trace R, shows the primary current with peak and same-polarity switching. The vertical scale in **Figures 2** and **3** is 10A per division, and Trace 2 is the relay control voltage. The primary current in **Figure 3** causes some core saturation (note that the current is not bipolar), but the saturation is much lower than that in **Figure 2**. Trace 1 shows the reduced primary current with the use of peak and opposite-polarity switching. Note that transformer designs vary widely; some may favor particular phase angles.




FIGURE 1

A μ C-controlled smart switch prevents transformer-core saturation, thus averting system crashes and prolonging the life of power-supply relays.

Inrush current from power-supply filter capacitors is also an important design consideration. By using a resistor, an inrush device, or an inductive input filter in the secondary winding, you can reduce this inrush surge. Another solution is to soft-start the transformer by using a resistor in the primary to limit inrush and saturation currents to an acceptable level. After a brief delay, a second solid-state relay shunts the resistor. The Microchip 12C508 μ C uses its internal 4-MHz RC oscillator for all timing. The chip is simple, inexpensive, reliable, and well-suited for this application. For wide temperature variations, you can obtain more accurate

timing by using a 32-kHz crystal. You can download **Listing 1**, the source code for the μ C's operation, from EDN's Web site www.ednmag.com. At the registered-user area, go into the Software Center to download the file from DI-SIG, #2170.

You can use either zero-crossing or random relays, but the random type works better for transformers. Set Pin 4 high for zero-crossing relays and low for random-turn-on relays. The HCPL-3760 optocoupler determines the polarity and phase of the ac line. The coupler is configured as a near-zero detector. Its output is set to switch on at 50V ac and off at 25V ac.

LISTING 1—SOURCE CODE FOR TRANSFORMER SOLID-STATE-RELAY CONTROLLER

```
;
; ROBERT LINDSEY
; 1908 KENNY AVE, CHAMPAIGN, IL 61821
; 217 384-5900

; This is a smart switch for controlling a solid state relay that is used
; for controlling the AC primary power to a large transformer. Transformers
; are noted for having extremely high inrush currents due to momentary
; core saturation if the polarization is correct. This version has an optional
; soft start switch that uses a resistor in the primary to limit current
; from the power supply filter capacitors and core saturation. The primary
; function of the code is to always turn the transformer on on the positive
; half of the ac cycle and turn it off on the negative half of the ac cycle.
; Random turn on solid state relays or zero crossing relays can be used.

title "Transformer SSR Controller"
list p=pic12C508, st=off, x=on, n=75, r=dec
include "pl2c508.inc"
__CONFIG B'000000001010'

;----- RAM REGISTERS -----
cblock H'0007'
COUNT1 ;test counter
COUNT2 ;test counter
endc

;----- PORT PIN ASSIGNMENTS -----
#define SSR1 GPIO,0 ;pin 7, output, SS relay 1
#define SSR2 GPIO,1 ;pin 6, output, SS relay 2
#define AC GPIO,2 ;pin 5, input, AC polarity, HCPL-3760
#define TE GPIO,3 ;pin 4, input, transformer enable
#define BLEEDER GPIO,4 ;pin 3, output, bleeder resistor
#define ZR GPIO,5 ;pin 2, input, zero crossing or random
;1=Zero crossing, 0=Random turn on

;-----
org 0
movlw 0
movwf OSCCAL ;int RC oscillator calibration value

movlw B'10000110' ;wake-up off, pull-ups on, T0 int clk,
option ;T0 prescaler, =128

;----- MAIN -----
bcf SSR1 ;set pin 7 latch low
bcf SSR2 ;set pin 6 latch low
bcf BLEEDER ;set pin 3 latch low

main movlw B'00101100' ;GPIO, 0 = output pin
tris GPIO ;set I/O pin functions

bcf SSR2 ;ssr2 is off
bsf BLEEDER ;bleeder is on
btfss TE ;is TE enable signal high
goto TX0 ;no, so keep checking

movlw 50 ;yes, make sure it was not a glitch
movwf COUNT2
call wait ;wait 50ms
btfss TE ;is TE enable still high
goto TX0 ;no, it was a glitch, keep waiting

bcf BLEEDER ;yes, turn off bleeder resistor
clrf COUNT2 ; 1/4 second delay
call wait ;

movlw 4 ;load COUNT2 with 4ms wait after zero crossing
movwf COUNT2 ;which will be near AC peak
call ACTrig ;wait for +zero crossing of AC voltage
call wait

bsf SSR1 ;turn on SS relay 1 (soft start resistor)
clrf COUNT2 ;load ms counter for 1/4 second delay
call wait ;allow power supply caps to charge

movlw 4 ;load COUNT2 with 4ms wait after zero crossing
movwf COUNT2 ;which will be near AC peak
call ACTrig ;wait for +zero crossing of AC voltage
call wait
bsf SSR2 ;turn on SS relay 2 (main)

;----- TURN OFF -----
; turn off transformer at negative peak

TX1 bsf SSR1 ;ssr1 is on
bsf SSR2 ;ssr2 is on
bcf BLEEDER ;bleeder is off
btfsc TE ;is TE enable signal low
goto TX1 ;no, it is still high, so keep checking

movlw 50 ;yes, make sure it was not a glitch
movwf COUNT2
call wait ;wait 50ms
btfsc TE ;is TE enable still low
goto TX1 ;no, it was a glitch, keep waiting

movlw 12 ;yes, load 12ms wait after zero crossing
btfsc ZR ;using zero crossing SS relay? ZR=1?
movlw 4 ;yes, load 4ms wait after zero crossing
movwf COUNT2 ;which will be near AC peak
call ACTrig ;wait for +zero crossing of AC voltage
call wait
bcf SSR2 ;turn off solid state relay 2
clrf COUNT2 ;1/4 second delay
call wait ;
movlw 12 ;load 12ms wait after zero crossing
btfsc ZR ;using zero crossing SS relay? ZR=1?
movlw 4 ;yes, load 4ms wait after zero crossing

movwf COUNT2 ;which will be near AC peak
call ACTrig ;wait for +zero crossing of AC voltage
call wait
bcf SSR1 ;turn off SS relay 1

clrf COUNT2 ;1/4 second delay
call wait ;

bsf BLEEDER ;turn on bleeder resistor
clrf COUNT2 ;1/4 second delay for power supply bleed down
call wait ;
goto main ;wait for turn-on

;----- AC TRIGGER -----
;wait for a low to high transition from the HCPL-3760 opto-coupler

ACTrig:
acl btfsc AC ;is AC input signal low
goto acl ;no, it is high, keep waiting until low
nop ;yes, it is low now


ac0 btfss AC ;is AC input signal high
goto ac0 ;no, it is low, keep waiting until high
nop ;yes, it is high now
return

;----- MS WAIT DELAY -----
;enter with milli-second value in COUNT2 register
;exits with COUNT1 and COUNT2 =0

wait:
wait1 clrf COUNT1
wait2 nop
decfsz COUNT1,1
goto wait2
decfsz COUNT2,1
goto wait1
return


end
```

FIGURE 2



Using only a zero-crossing relay results in core saturation and a disastrous 46A peak current in the transformer's primary winding.

FIGURE 3



The smart-switch circuit in Figure 1 greatly reduces core saturation, resulting in well-behaved primary current. Trace R1 results from peak and same-polarity switching; trace 1 represents peak and opposite-polarity switching.

One internal diode in the optocoupler rectifies the ac signal to indicate the positive half cycle. The μC has two solid-state-relay outputs: SSR1 and SSR2. When the Transformer Enable input goes high, the μC waits 250 msec, detects the next positive edge from the optocoupler, waits 12 msec, and then turns off SSR1. SSR2 has a 250-msec delay from SSR1 and operates as a last-on, first-off output to shunt a soft-start resistor. Pin 3 is an optional output for a power-supply bleed-off switch or a status indicator. (DI #2170) **EDN**